NAG Library Routine Document

C02AHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C02AHF determines the roots of a quadratic equation with complex coefficients.

2 Specification

```
SUBROUTINE C02AHF(AR, AI, BR, BI, CR, CI, ZSM, ZLG, IFAIL)
INTEGER            IFAIL
double precision   AR, AI, BR, BI, CR, CI, ZSM(2), ZLG(2)
```

3 Description

C02AHF attempts to find the roots of the quadratic equation $az^2 + bz + c = 0$ (where a, b and c are complex coefficients), by carefully evaluating the 'standard' closed formula

$$z = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}.$$

It is based on the routine CQDRTC from Smith (1967).

Note: it is not necessary to scale the coefficients prior to calling the routine.

4 References

Smith B T (1967) ZERPOL: A zero finding algorithm for polynomials using Laguerre's method *Technical Report* Department of Computer Science, University of Toronto, Canada

5 Parameters

1: AR – ***double precision*** *Input*
2: AI – ***double precision*** *Input*

On entry: AR and AI must contain the real and imaginary parts respectively of a, the coefficient of z^2.

3: BR – ***double precision*** *Input*
4: BI – ***double precision*** *Input*

On entry: BR and BI must contain the real and imaginary parts respectively of b, the coefficient of z.

5: CR – ***double precision*** *Input*
6: CI – ***double precision*** *Input*

On entry: CR and CI must contain the real and imaginary parts respectively of c, the constant coefficient.

7: ZSM(2) – ***double precision*** array *Output*

On exit: the real and imaginary parts of the smallest root in magnitude are stored in ZSM(1) and ZSM(2) respectively.

8: ZLG(2) – ***double precision*** array *Output*

On exit: the real and imaginary parts of the largest root in magnitude are stored in ZLG(1) and ZLG(2) respectively.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, $(\mathrm{AR}, \mathrm{AI}) = (0, 0)$. In this case, ZSM(1) and ZSM(2) contain the real and imaginary parts respectively of the root $-c/b$.

IFAIL $= 2$

On entry, $(\mathrm{AR}, \mathrm{AI}) = (0, 0)$ and $(\mathrm{BR}, \mathrm{BI}) = (0, 0)$. In this case, ZSM(1) contains the largest machine representable number (see X02ALF) and ZSM(2) contains zero.

IFAIL $= 3$

On entry, $(\mathrm{AR}, \mathrm{AI}) = (0, 0)$ and the root $-c/b$ overflows. In this case, ZSM(1) contains the largest machine representable number (see X02ALF) and ZSM(2) contains zero.

IFAIL $= 4$

On entry, $(\mathrm{CR}, \mathrm{CI}) = (0, 0)$ and the root $-b/a$ overflows. In this case, both ZSM(1) and ZSM(2) contain zero.

IFAIL $= 5$

On entry, $\tilde{b}$ is so large that $\tilde{b}^2$ is indistinguishable from $\tilde{b}^2 - 4\tilde{a}\tilde{c}$ and the root $-b/a$ overflows, where $\tilde{b} = \max(|\mathrm{BR}|, |\mathrm{BI}|)$, $\tilde{a} = \max(|\mathrm{AR}|, |\mathrm{AI}|)$ and $\tilde{c} = \max(|\mathrm{CR}|, |\mathrm{CI}|)$. In this case, ZSM(1) and ZSM(2) contain the real and imaginary parts respectively of the root $-c/b$.

If IFAIL > 0 on exit, then ZLG(1) contains the largest machine representable number (see X02ALF) and ZLG(2) contains zero.

7 Accuracy

If IFAIL $= 0$ on exit, then the computed roots should be accurate to within a small multiple of the ***machine precision*** except when underflow (or overflow) occurs, in which case the true roots are within a small multiple of the underflow (or overflow) threshold of the machine.

8 Further Comments

None.

9 Example

This example finds the roots of the quadratic equation $z^2 - (3.0 - 1.0i)z + (8.0 + 1.0i) = 0$.

9.1 Program Text

```
*     C02AHF Example Program Text
*     Mark 14 Release. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION AI, AR, BI, BR, CI, CR
      INTEGER          IFAIL
*     .. Local Arrays ..
      DOUBLE PRECISION ZLG(2), ZSM(2)
*     .. External Subroutines ..
      EXTERNAL         C02AHF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C02AHF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) AR, AI, BR, BI, CR, CI
      IFAIL = 1
*
      CALL C02AHF(AR,AI,BR,BI,CR,CI,ZSM,ZLG,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Roots of quadratic equation'
         WRITE (NOUT,*)
         WRITE (NOUT,99999) 'z = ', ZSM(1), ZSM(2), '*i'
         WRITE (NOUT,99999) 'z = ', ZLG(1), ZLG(2), '*i'
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,99998) ' ** C02AHF returned with IFAIL = ', IFAIL
      END IF
*
99999 FORMAT (1X,A,1P,E12.4,SP,E14.4,A)
99998 FORMAT (1X,A,I5)
      END
```

9.2 Program Data

```
C02AHF Example Program Data
 1.0    0.0   -3.0    1.0    8.0    1.0        :AR  AI  BR  BI  CR  CI
```

9.3 Program Results

```
 C02AHF Example Program Results

 Roots of quadratic equation

 z =   1.0000E+00   +2.0000E+00*i
 z =   2.0000E+00   -3.0000E+00*i
```